# Lecture 22

# Input/Output: Printing

Text: Chapter 20

Besides printing characters on the paper, the printer must interpret some control codes

- new page (Form Feed)
- line feed
- carriage return
- tabs

It must also communicate back to the processor

- busy
- out of paper
- off-line



# PRINT CHARACTERS

DOS INT 21h, Function 40h

AH Function 40h
BX File handle 04h
CX Number of Characters to Print
DX Address of text

EXAMPLE:
Print “Computer Science 201”

```
Message   DB  ‘Computer Science 201’,
          DB   0Dh,0Ah,0Ch  ;cr,lf,ff

          MOV  AH,40h
          MOV  BH,04
          MOV  CX,22
          LEA  DX,Message
          INT  21h
```

The message is terminated with

CR so the next line starts in column 1.
LF so the next line is printed on the next line down.
FF so the printer will print the page immediately

Some printers, especially laser printers, will buffer the characters until you request the page be ejected.

```
TITLE   P20PRTNM  Accept entered names and print
        .MODEL  SMALL  (.EXE)
        .STACK  64
; -----------------------------------------------
        .DATA
NAMEPAR LABEL   BYTE    ;Keyd parameter list:
MAXNLEN DB      20      ;  max length of name
NAMELEN DB      ?       ;  actual length entered
NAMEFLD DB      20 DUP(' ')  ;  name entered
        ;Heading line:
HEADG   DB      'List of Employee Names   Page  '
PAGECTR DB      '01', 0AH, 0AH

FFEED   DB      0CH     ;Form feed
LFEED   DB      0AH     ;Line feed
LINECTR DB      01
PROMPT  DB      'Name? '
; -----------------------------------------------
        .CODE
BEGIN   PROC    FAR
        MOV     AX,@data ;Initialize
        MOV     DS,AX    ;  segment
        MOV     ES,AX    ;  registers
        CALL    Q10CLR   ;Clear screen
        CALL    M10PAGE  ;Page heading
A20LOOP:
        MOV     DX,0000  ;Set cursor to 00,00
        CALL    Q20CURS
        CALL    D10INPT  ;Provide input of name
        CALL    Q10CLR
        CMP     NAMELEN,00  ;No name entered?
        JE      A30      ;  no name, exit
        CALL    E10PRNT  ;  name, prep printing
        JMP     A20LOOP
A30:
        MOV     CX,01    ;End of processing:
        LEA     DX,FFEED ;  one character
        CALL    P10OUT   ;  for form feed,
```

```
        MOV     AX,4C00H ;  exit to DOS
        INT     21H
BEGIN   ENDP
;       Accept input of name:
;       ---------------------
D10INPT PROC    NEAR
        MOV     AH,40H   ;Request display
        MOV     BX,01    ;
        MOV     CX,05    ;  5 characters
        LEA     DX,PROMPT ;  prompt message
        INT     21H
        MOV     AH,0AH   ;Request keyboard
        LEA     DX,NAMEPAR;  input
        INT     21H
        RET
D10INPT ENDP
;       Prepare for printing:
;       ---------------------
E10PRNT PROC    NEAR
        CMP     LINECTR,60 ;End of page?
        JB      E20      ;  no  - bypass
        CALL    M10PAGE  ;  yes - print heading
E20:
        MOV     CH,00
        MOV     CL,NAMELEN ;Set no. of characters
        LEA     DX,NAMEFLD ;Set address of name
        CALL    P10OUT   ;Print name
        MOV     CX,01    ;One
        LEA     DX,LFEED ;  line feed
        CALL    P10OUT
        INC     LINECTR  ;Add to line count
        RET
E10PRNT ENDP
;       Page heading routine:
;       ---------------------
M10PAGE PROC    NEAR
        CMP     WORD PTR PAGECTR,3130H ;1st page?
        JE      M30      ;  yes - bypass
```

```
        MOV     CX,01    ;
        LEA     DX,FFEED ;  no --
        CALL    P10OUT   ;  form feed,
        MOV     LINECTR,03 ;  reset line count
M30:
        MOV     CX,36    ;Length of heading
        LEA     DX,HEADG ;Address of heading
M40:
        CALL    P10OUT
        INC     PAGECTR+1 ;Add to page count
        CMP     PAGECTR+1,3AH ;Page no. = hex 3A?
        JNE     M50      ;  no  - bypass
        MOV     PAGECTR+1,30H ; yes- set to ASCII
        INC     PAGECTR  ;
M50:    RET
M10PAGE ENDP
;       Print routine:
;       --------------
P10OUT  PROC    NEAR     ;CX and DX set on entry
        MOV     AH,40H   ;Request print
        MOV     BX,04    ;Handle
        INT     21H
        RET
P10OUT  ENDP
;       Clear screen:
;       -------------
Q10CLR  PROC    NEAR
        MOV     AX,0600H ;Request scroll
        MOV     BH,60H   ;Attribute
        MOV     CX,0000  ;From 00,00
        MOV     DX,184FH ;  to 24,79
        INT     10H
        RET
Q10CLR  ENDP
;       Set cursor row/col:
;       -------------------
Q20CURS PROC    NEAR     ;DX set on entry
        MOV     AH,02H   ;Request set cursor
```

```
         MOV     BH,00    ;Page number 0
         INT     10H
         RET
Q20CURS  ENDP
         END     BEGIN
```

## Other printer control codes

08h Backspace
0Bh Vertical tab
0Fh Condensed Mode ON
12h Condensed Mode OFF
0Eh Expanded Mode ON
14h Expanded Mode OFF

These must be preceded by the ESC character (1Bh)

1B 30h Set line spacing 8 lines per inch
1B 32h Set line spacing 6 lines per inch
1B 45h Set emphasized printing ON
1B 46h Set emphasized printing OFF

# BIOS INT 17h

02h Determine the printer's status

AH bits:
- 0 Time out
- 3 Input/Output error
- 4 Selected
- 5 Out of Paper
- 6 Acknowledged from printer
- 7 Not busy

01h Initialize Printer Port

(Sends FF character)

00h Print a character

(Also returns printer status to AH)

EXAMPLE:

Make the screen red until the printer is ready. Then make it green and wait for the user to strike any key.

```
     include c:\bp\bin\cs201\clearscr.lib
     include c:\bp\bin\cs201\getkey.lib

<stack segment>
<data segment>
;
again:
     mov  ah,02h  ;read port
     mov  dx,00   ;select lpt1:
     int 17h      ;ask for status
     test   ah,00101001b   ;ready?
     jz   ready   ;yes- make screen green
     clearscr   43h ;no- make screen red
     jmp   again  ;ask status again

; printer ready- set green

ready:
     ClearScr  23h  ; set screen green
     getkey         ;user hits any key

exit:
        MOV      AX,4C00H
        INT      21H
BEGIN   ENDP
CODESG  ENDS
         END      BEGIN
```

# Exercises - Lecture 22

1. Adapt the program on page 22.8 so that is prints a message if the printer is out of paper.